
	Other comments:
	


4. Corrections required for TC 22.4.7
Change 1 – Correction to function f_NBIOT_508Check_CP_MONonSMSDataTransfer
	Function name
	f_NBIOT_508Check_CP_MONonSMSDataTransfer

	Reason for change
	1. According to TS 36.508 section 8.1.5A.3.3, the steps 3a1 to 3a4 should be executed only if px_DoAttachWithoutPDN is set to TRUE. If this pixit is seto FALSE, then Step 3b1 should be executed. The TTCN implementation does not follow this procedure sending a ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST at Step 3a2 independent of whether px_DoAttachWithoutPDN is set to TRUE or FALSE and waiting to receive an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT at Step 3a3. This is incorrect
2. The ESM Data Transport message would be received as Piggy Backed message in the CONTROL PLANE SERVICE REQUEST. However, the TTCN does not have code to handle this.

	Summary of change
	1. Brought the TTCN code from 3a1 to 3a4 under the purview of the pixit px_DoAttachWithoutPDN
2. The template cr_NAS_IndicationWithPiggybacking needs to be used for handling the piggy backed ESM DATA TRANSPORT message.

	TTCN module
	NBIOT_CommonProcedures.ttcn

	MCC160 Comment
	Not Needed.


Before change
	function f_NBIOT_508Check_CP_MONonSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                    template (present) UserDataContainer p_DataExpected,

                                                    template (present) EstablishmentCause_NB_r13 p_EstCause := mo_Data) runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var integer v_PDNIndex := 0;

    var template (value) PDN_Address v_PDN_Address;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) {

      v_AttachWithoutPDN := true_;

      v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                        v_PDNType,

                                                                        -,

                                                                        v_HeaderCompressConfigRX);

      v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B))));                // or without PDN Connectivity

    } else {

      v_NasInd_Expected := cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), cr_ESM_DATA_TRANSPORT (?, p_DataExpected));

    }

    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def (p_CellId);

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    if (ispresent (v_NasInd_Rxd.Pdu.PiggybackedPduList)) { // Step 3a1a1

      if (ischosen (v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST)) {

        v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1a1");

      } else {

        f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Unexpected Message at Step3a1a1");

      }

    } else { // Step 3a1b1

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b1");

      // Step 3a1b2

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;

      v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b2");

    }

    v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

    f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

    v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

    v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

    if (isvalue(v_PcoFromUE)) {

        f_NBIOT_MobileInfo_SetExtPCO (v_PDNIndex, valueof(v_PcoFromUE));

    }

    if (ispresent (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

      v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

      v_APN.iei := omit;

    } else {

      v_APN := valueof(cs_AccessPointName(px_AccessPointName));

    }

    f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

    if (match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) {

      v_PDN_Address := cs_PDN_Address_NonIP;

    } else {

      v_PDN_Address := f_NBIOT_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE));

    }

    // Step 3a2

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                            f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                            v_APN,

                                                                                            v_PDN_Address,

                                                                                            cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                            -,

                                                                                            -,

                                                                                            -,

                                                                                            v_HeaderCompressConfigTX,

                                                                                            cs_ControlPlaneInd,

                                                                                            cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

    // Step 3a3

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a3");

    if (pc_IP_PDN) {

      f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

    }

    if (px_DoAttachWithoutPDN) {

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cr_ESM_DATA_TRANSPORT (?, p_DataExpected))));

    }

    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

  }




After change
	function f_NBIOT_508Check_CP_MONonSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                    template (present) UserDataContainer p_DataExpected,

                                                    template (present) EstablishmentCause_NB_r13 p_EstCause := mo_Data) runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var integer v_PDNIndex := 0;

    var template (value) PDN_Address v_PDN_Address;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    var boolean v_ConfigurePDN := false;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) { //WA#
      v_AttachWithoutPDN := true_;

      v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                        v_PDNType,

                                                                        -,

                                                                        v_HeaderCompressConfigRX);

      v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                            cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B))));                // or without PDN Connectivity 

    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def();

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    if (ispresent (v_NasInd_Rxd.Pdu.PiggybackedPduList)) {

      if (ischosen (v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.eSM_DATA_TRANSPORT)) { // Step 3b1

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3b1");

      }

      else if (ischosen (v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST)) { // Step 3a1a1 /*@sic R5s170488 sic@*/ 

        v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

        v_ConfigurePDN := true; // @sic R5s170364 sic@

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1a1");

      } else {

        f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Unexpected Message at Step3a1a1");

      }

    } else { // Step 3a1b1

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b1");

      // Step 3a1b2

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;

      v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

      v_ConfigurePDN := true; // @sic R5s170364 sic@

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b2");

    }

    if (v_ConfigurePDN) { // @sic R5s170364 sic@

      v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

      f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                               v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

      v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

      v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

      if (isvalue(v_PcoFromUE)) {

        f_NBIOT_MobileInfo_SetExtPCO(v_PDNIndex, valueof(v_PcoFromUE));

      }

      if (ispresent(v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

        v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

        v_APN.iei := omit;

      } else {

        v_APN := valueof(cs_AccessPointName(px_AccessPointName));

      }

      f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

      if (match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) {

        v_PDN_Address := cs_PDN_Address_NonIP;

      } else {

        v_PDN_Address := f_NBIOT_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE));

      }

      // Step 3a2

      SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                     tsc_SRB1bis,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                               f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                               v_APN,

                                                                                               v_PDN_Address,

                                                                                               cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                               -,

                                                                                               -,

                                                                                               -,

                                                                                               v_HeaderCompressConfigTX,

                                                                                               cs_ControlPlaneInd,

                                                                                               cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

      // Step 3a3

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a3");

      if (pc_IP_PDN) {

        f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

      }

      // Step 3a4

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cr_ESM_DATA_TRANSPORT (?, p_DataExpected))));

     }

    } else { //WA#

     v_NasInd_Expected := cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected_PartiallyCiphered,cr_CONTROL_PLANE_SERVICE_REQUEST(?,cr_DataServiceType(-, '000'B)), cr_ESM_DATA_TRANSPORT (?, p_DataExpected));

     f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

     f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

     // Step 2

     f_NBIOT_RRC_ConnectionSetup_Def ();

     // Step 3

     v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    }
    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

  }



Change 2 – Correction to function ‘f_TC_22_4_7”
	Function name
	f_TC_22_4_7

	Reason for change
	The local end config to configure Msg4 with RRC Connection Reject to be sent at Step 6 needs to be done a little earlier because in function f_NBIOT_RRC_ConnectionRelease_Common, there is the option to receive an RRC Connection Request and if the UE does send an RRC Connection Request message it would result in the default RRC Connection Setup message being sent to the UE rather than the RRC Connection Reject message.

	Summary of change
	It is recommended to call the f_NBIOT_SS_ConfigRachProcedureMsg4 for RRC Connection Reject before the f_NBIOT_RRC_ConnectionRelease and f_Delay so that there is enough time to respond to the incoming RRC Connection Request with the RRC Connection Reject.

	TTCN module
	NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	Will be implemented according to outcome of draft prose CR associated with R5s170562


Before change
	<<SKIPPED CODE>>
//@siclog "Step 3C" siclog@

    //@sic R5-171547 sic@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

    //@siclog "Step 3D" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 3E" siclog@

    //@sic R5-171547 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //Configure Msg4 with RRCConnectionReject for step 6

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, cs_RRCConnectionReject_NB(v_XtdWaitTime));
<<SKIPPED CODE>>


After change
	<<SKIPPED CODE>>
 //@siclog "Step 3C" siclog@

 //@sic R5-171547 sic@

 //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, cs_UserDataContainer(v_UserData)))));//@sic R5s170364 sic@

   //Configure Msg4 with RRCConnectionReject for step 6 
   f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, cs_RRCConnectionReject_NB(v_XtdWaitTime)); //WA#
    //@siclog "Step 3D" siclog@

    //@sic R5-171547 sic@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 3E" siclog@

    //@sic R5-171547 sic@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);
<<SKIPPED CODE>>


